
1 Introduction and tutorial

1.1 Heptagon: short presentation
Heptagon is a synchronous dataflow language, with a syntax allowing the expression of control
structures (e.g., switch or mode automata).

A typical Heptagon program will take as input a sequence of values, and will output a sequence
of values. Then, variables (inputs, outputs or locals) as well as constants are actually variable
or constant streams. The usual operators (e.g., arithmetic or Boolean operators) are applied
pointwise on these sequences of values.

For example, the Heptagon program below is composed of one node plus, performing the
pointwise sum of its two integer inputs:

node plus(x:int,y:int) returns (z:int)
let
z = x + y;
tel

x and y are the inputs of the node plus; z is the output. x, y and z are of type int, denoting
integer streams. z is defined by the equation z =x + y.

An execution of the node plus can then be:

x 1 2 3 4 . . .
y 1 2 1 2 . . .

plus(x, y) 2 4 4 6 . . .

1.2 Compilation
The Heptagon compiler is named heptc. Its list of options is available by :

> heptc -help

Every options described below are cumulable.
Assuming that the program to compile is in a file named example.ept, then one can compile

it by typing :

> heptc example.ept

However, such compilation will only perform standard analysis (such as typing, causality,
scheduling) and output intermediate object code, but not any final or executable code.

The Heptagon compiler can thus generate code in some general languages, in order to obtain
either a standalone executable, or a linkable library. The target language must then be given by
the -target option:

> heptc -target <language> example.ept

Where <language> is the name of the target language. For now, available languages are C
(c option) and Java (java option).

1.3 Generated code
The generic generated code consists, for each node, of two imperative functions:

• one “reset” function, used to reset the internal memory of the node;

• one “step” function, taking as input the nodes inputs, and whose call performs one step of
the node, updates the memory, and outputs the nodes outputs.

A standard way to execute Heptagon program is to compile the generated files together with
a main program of the following scheme :

call the reset function
for each instant
get the inputs values
outputs ← step(inputs)
do something with outputs values

Appendix A give specific technical details for each target language.

1.4 Simulation
A graphical simulator is available: hepts. It allows the user to simulate one node by providing a
graphical window, where simulation steps can be performed by providing inputs of the simulated
node.

This simulator tool interacts with an executable, typically issued of Heptagon programs
compilation, and which await on the standard input the list of the simulated node’s inputs, and
prints its outputs on the standard output. Such executable, for the simulation of the node f,
can be obtained by the -s <node> option:

> heptc -target c -s f example.ept

We can then directly compile the generated C program (whose main function stand in the
_main.c file):

> cd example_c
> gcc -Wall -c example.c
> gcc -Wall -c _main.c
> gcc -o f_sim _main.o example.o # executable creation

This executable f_sim can then be used with the graphical simulator hepts, which takes as
argument:

• The name of the module (capitalized name of the program without the .ept extension),

• the name of the simulated node,

• the path to the executable f_sim.

> hepts -mod Example -node f -exec example_c/f_sim

2 Syntax and informal semantics
Heptagon programs are synchronous Moore machines, with parallel and hierarchical composition.
The states of such machines define dataflow equations. The Figure 1 gives an example of such
program.

A

y = false

C
y = true

Idle y1 = false

Act y1 = true

cc

Idle y2 = false

Act y2 = true

dd

y = y1 ∧ y2

c

c ∧ d

d

Figure 1: Mixed state and dataflow example

2.1 Nodes
Heptagon programs are structured in nodes: a program is a sequence of nodes. A node is a
subprogram with a name f , inputs x1, . . . , xn, outputs y1, . . . , yp, local variables z1, . . . , zq and
declarations D. yi and zi variables are to be defined in D, using operations between values of
xj , yj , zj . Figure 2 gives the syntax of node definitions, together with a graphical syntax used
in this manual1. The declaration of one variable comes with its type (ti, t′i and t′′i being the type
of respectively xi, yi and zi).

The program of the Figure 1 can thus be structured as the semantically equivalent program
of the Figure 3. The Figure 4 gives the textual syntax of this program.

Heptagon allows to distinguish, by mean of clocks and control structures (switch, automata),
for declarations and expressions, the discrete instants of activation, when the declarations and
expressions are computed and progress toward further states, and other instants when neither
computation nor progression are performed.

1declaration of local variables are mandatory for the compiler in the textual syntax, however we will sometimes
omit it in the graphical syntax for the sake of brevity

f(x1 : t1, . . . , xn : tn) = y1 : t′1, . . . , yp : t′p

D

node f(x1:t1;. . .;xn:tn) returns (y1:t′1,. . .,yp:t
′
p)

var z1:t′′1,. . .,zq:t
′′
q;

let
D
tel

Figure 2: Graphical and textual syntax of node definition

2.2 Expressions
2.2.1 Values and combinatorial operations

Heptagon is a dataflow language, i.e., every value, variable or constant, is actually a stream of
value. The usual operators (e.g., arithmetic or Boolean operators) are applied pointwise on these
sequences of values, as combinatorial operations (as opposed to sequential operations, taking into
account the current state of the program: see delays in Section 2.2.2).

Thus, x denotes the stream x1.x2. . . ., and x + y is the stream defined by (x + y)i = xi+ yi.

x x1 x2 x3 x4 . . .
y y1 y2 y3 y4 . . .

x+ y x1 + y1 x2 + y2 x3 + y3 x4 + y4 . . .

2.2.2 Delays

Delays are the way to introduce some state in a Heptagon program.

• pre x gives the value of x at the preceding instant. The value at the first instant is
undefined.

• x ->y takes the value of x at the first instant, and then the value of y;

• x fby y is equivalent to x ->pre y.

x x1 x2 x3

y y1 y2 y3
pre x ⊥ x1 x2

x ->y x1 y2 y3
x fby y x1 y1 y2

Here is a small example of a node that sums its inputs:

node sum(i:int) returns (o:int)
let
o = 0 fby (o + i)
tel

4



h(a, b) = y

Idle y = false

Act y = true

a ∧ ba

g(a, b) = y

y1 = h(a, c)
y2 = h(b, d)
y = y1 ∧ y2

f(c, d) = y

A

y = false

C
y = true

B y = g(c, d)

c

c ∧ d

d

Figure 3: Structured program example

node h(a:bool) returns (y:bool)
let
automaton
state Idle
do y = false
until a then Active
state Active
do y = true
until a then Idle

end
tel

node g (a,b:bool) returns (y:bool)
var y1,y2 : bool;
let
y = y1 & y2;
y1 = h(a);
y2 = h(b);
tel

node f (c,d:bool) returns (y:bool)
let
automaton
state A
do y = false
until c then B
state B
do y = g(c,d)
until c & d then C
state C
do y = true
until d then A

end
tel

Figure 4: Textual syntax

2.2.3 Clocks

It is possible to mix streams with different rates by having streams that are not present at
each instant. This can be done using the when operator, that samples a stream according to
a condition, either a boolean or an enumerated value. If x is a stream that is always present,
then x when c (resp x when C(y)) is a stream equal to x but only present when c is true (resp.
y =C). x whenot c is a shortcut for x when (not c). The split operator allows to sample
a stream according to all possible values, i.e. split c (x) =x when c, x whenot c if x is a
boolean and split y (x) =x when C1(y), .., x when Cp(y) if y has an enumerated type
defined by type t =C1 | ... | Cp.

x x1 x2 x3

c true false true
x when c x1 . x3

y C ′ C C ′

x when C(y) . x2 .

The clock of an expression (resp. a stream) is a boolean stream that defines the instants
when it is present. The clock of the streams that are always present is called base or .. If x has
clock ck, denoted x :: ck, then x when c has clock ck on c.

The merge operator joins slow streams to create a faster stream.

c true false true
x x1 . x3

y . y2 .
merge c x y x1 y2 x3

merge expects complementary streams. If z =merge c x y and z :: ck, then we must have
x :: ck on c and y :: ck onot c. It is thus different from if c then x else y that expects
all its arguments to have the same clock. An analysis pass called clock calculus checks that these
conditions are met.

Here is a first example of a bidirectional counter:

type modes = Up | Down

node two(m:modes;v:int) returns (o:int)
var x:int;

x_up:int :: . on Up(m);
x_down:int :: . on Down(m);

let
o = 0 fby x;
x_up = o when Up(m) + 1;
x_down = o when Down(m) + 1;
x = merge m (Up -> x_up) (Down -> x_down)
tel

Note that clocks are inferred by the compiler, so the clocking annotations are optional.
It is important to understand the interaction between sampling and delays. The value of a

delay only changes when it is activated, that is, when its clock is true. As soon as a function f
contains some delay operator, sampling its inputs is not equivalent to sampling its outputs, that

is, f(x when c) 6= (f x) when c.

c true false true false
counter(1) 0 1 2 3

counter(1) when c 0 . 2 .
counter(1 when c) 0 . 1 .

2.3 Declarations
A declaration D can be either :

• an equation x = e, defining variable x by the expression e at each activation instants ;

• a node application (y1, . . . , yp) = f(e1, . . . , en), defining variables y1, . . . , yp by application
of the node f with values e1, . . . , en at each activation instants ;

• parallel declarations of D1 and D2, noted graphically D1

...D2 and textually D1;D2. Vari-
ables defined in D1 and D2 must be exclusive. The activation of this parallel declaration
activate both D1 and D2, which are both computed and both progress ;

• a switch control structure ;

• an automaton.

2.3.1 Switch control structures

The switch control structure allows to controls which equations are evaluated:

type modes = Up | Down

node two(m:modes;v:int) returns (o:int)
var last x:int = 0;
let
o = x;
switch m
| Up do x = last x + v
| Down do x = last x - v
end
tel

The last keyword defines a memory which is shared by the different modes. Thus, last x
is the value of the variable x in the previous instant, whichever was the activated mode.

2.3.2 Automata

An automaton is a set of states (one of which being the initial one), and transitions between
these states, triggered by Boolean expressions. A declaration is associated to each state. The
set of variables defined by the automaton is the union, not necessarily disjoint (variables can
have different definitions in different states, and can be partially defined : in this case, when the
variable is not defined in an active state, the previous value of this variable is taken.

At each automaton activation instant, one and only one state of this automaton is active (the
initial one at the first activation instant). The declaration associated to this active state is itself
activated and progress in this activation instant.

Example The following example gives the node updown. This node is defined by an automaton
composed of two states:

• the state Up gives to x its previous value augmented of 1

• the state Down gives to x its previous value diminued of 1

This automaton comprises two transitions:

• it goes from Up (the initial state) to Down when x becomes greater or equal than 10;

• it goes from Down to Up when x becomes less or equal 0.

node updown() returns (y:int)
var last x:int = 0;
let
y = x;
automaton
state Up
do x = last x + 1
until x >= 10 then Down
state Down
do x = last x - 1
until x <= 0 then Up

end
tel

current state Up Up Up Up Up Up Up Up Up Up Down Down Down . . .
y 1 2 3 4 5 6 7 8 9 10 9 8 7 . . .

Expressions on outgoing transitions of this active state are evaluated, so as to compute the
next active state : these are weak transitions. Transitions are evaluated in declaration order, in
the textual syntax. If no transition can be triggered, then the current state is the next active
state.

2.4 Structured types
2.4.1 Arrays

Arrays are denoted using the ^ symbol. For instance, int^10 is the type of arrays of size 10
containing integers. Arrays can be multidimensional, like int^3^2. One should note that indices
appear in reverse order compared to C: int^3^2 should be understood as (int^3)^2, that is an
array of size 2 containing arrays of size 3 (this would have been written int t[2][3] in C).

The following operators are defined for arrays:

• Accessing an element at a constant index: t[4]

• Accessing an element at a dynamic index with a default value: z =t.[x] default v is a
stream defined by zi = ti[xi] if 0 ≤ xi < n and zi = vi otherwise, where n is the size of t.

• Accessing an element at a truncated index: z =t[>x<] is defined by zi = ti[min(max(0, xi), n−
1)], where n is the size of t.

• Modifying an element: t’ =[ t with [x] =v ] is a new array equal to t, except for the
element at index x which is equal to v, that is for all j in [0, n − 1] such that j 6= xi,
t′i[j] = t[j] and t′i[xi] = vi.

• Defining an array by copying one value: x^n is the array of size n whose elements are all
equal to x.

• Defining an array explicitely: [1, x, 3, y, 5].

• Extracting a slice: t[n..m] returns the sub array of t of size m-n+1 starting at index n
and ending at index m.

• Concatening arrays: t1@t2.

It is also possible to use iterators to operate on arrays. For instance, one can add two
arrays pointwise by doing t =map<<n>> (+)(t1, t2) (where n is the size of t1, t2 and t). The
following iterators are defined:

• map applies a node pointwise to the input arrays and outputs one or several arrays. If f has
type t1 * t2 ->t’1 * t’2 then map<<n>> f has type t1^n * t2^n ->t’1^n * t’2^n.

• mapi is the same as map but the iterated function should expect another integer argu-
ment, which will be equal to the index of the inputs in the input arrays. If f has type
t1 * t2 * int ->t’1 * t’2 then map<<n>> f has type t1^n * t2^n ->t’1^n * t’2^n.

• fold iterates a node on an array accumulating values. If f has type t1 * t ->t then
fold<<n>> f has type t1^n * t ->t. For instance, fold<<2>> f(t, 0) is equivalent to
f(t[1], f(t[0], 0)).

• foldi is to fold what mapi is to map.

• mapfold combines the result of map and fold, by accumulating a value and outputting a new
array. If f has type t1 * t ->t’1 * t then mapfold<<n>> f has type t1^n * t ->t’1^n * t.

T T’

N

FIG. 4.5 – L’opération map.

Nelt_in elt_out

FIG. 4.6 – Le nœud itéré.

node N(elt_in : int) returns (elt_out : int);
let

elt_out = elt_in -> elt_in + pre(elt_out);
tel

L’itération de ce nœud grâce à unmap à l’intérieur d’un nœud test (ci-dessous) permet de calculer le
tableau T’ tel que ∀i ∈ [0..n− 1]. T �[i] = N(T [i]).

node test(T : intˆn) returns (T’ : intˆn);
let

T’ = map�N;n�(T);
tel

4.3.1.2 Code séquentiel correspondant

Le code que l’on produit pour l’équation T’ = map�N,n�(T) est :

for(i=0;i<n;i++){
T’[i] = N(T[i]);

}

4.3.2 red

La réduction, fréquemment appelée foldl ou foldr [GLP93], permet de calculer un accumulateur en
parcourant un ou plusieurs tableau(x). Soit la fonction g = λt, acc.acc�. La réduction res d’un tableau
T à l’aide de cette fonction peut être calculée de la manière suivante :

res = red(init, T, g),

où init est une expression d’initialisation de la réduction. En , on utilise l’opérateur red qui a
la syntaxe suivante. Si N est un nœud de profil :

τ × τ1 × τ2 × . . .× τl → τ �

et si n est une constante statique, alors red�N;n� est un nœud de profil :

τ × τ1ˆn×τ2ˆn× . . .× τlˆn→ τ �

Le schéma 4.7 montre un red simple, avec un tableau en entrée. Le schéma 4.8 montre plus en détail
le nœud itéré.

init N

res
T

FIG. 4.7 – L’opération de réduction.

acc_in

acc_out

elt N

FIG. 4.8 – Le nœud itéré.

4.3.2.1 Exemple

Voici un nœud qui calcule la somme de ses 2 entrées entières:

node plus(acc_in, elt : int) returns (acc_out : int);
let
acc_out = acc_in + elt;

tel

L’itération de ce nœud avec l’opérateur red permet de calculer la somme des éléments d’un tableau :

node redPlus(T : intˆn) returns (res : int’);
let

res = red�plus;n�(0,T);
tel

4.3.2.2 Code séquentiel correspondant

Le code que l’on produit pour l’équation T’ = red�plus,n�(0,T) est :

int res = 0;
for(i=0;i<n;i++){
res = plus(res,T[i]);

}

Remarque 6 (Concernant le nœud itéré) Un nœud itéré par un red aura toujours la structure sui-
vante :

• Au moins 2 entrées : la première correspondant à l’accumulateur, les suivantes correspondant
aux éléments des tableaux donnés en paramètre au red ;

• Toujours une seule sortie : celle correspondant à l’accumulateur.
D’autre part, on doit s’assurer que les horloges des accumulateurs de sortie et d’entrée du nœud itéré
sont compatibles puisque ces variables se retrouvent « branchées » entre deux instances du nœud
itéré.

L’unicité de l’accumulateur ne limite en rien les possibilités offertes : en effet, on peut utiliser des
accumulateurs à type structuré, avec autant de champs que désiré. Nous verrons cela dans les exemples
en 4.3.5.

(b) fold

init N

T’T
res

FIG. 4.11 – L’opération map_red.

N

accu_out

acc_in

elt_outelt_in

FIG. 4.12 – Détail.

node test(init : int ; T : intˆn)
returns (res : int ; T’ : intˆn);
let
res,T’ = map_red�sumCumul;n�(init,T);

tel

4.3.4.2 Code séquentiel correspondant

Le code que l’on produit pour l’équation T’ = map_red�N,n�(T) est le suivant :

acc_out = acc_in;
for(i=0;i<n;i++){
acc_out,T’[i] = N(acc_out,T[i]);

}

Remarque 8 (Concernant le nœud itéré) Au moins 2 entrées et 2 sorties :
• la première entrée et la première sortie correspondent à l’accumulateur ;
• les autres entrées/sorties correspondent aux éléments des tableaux consommés/produits par l’ité-
ration.

Les contraintes d’horloges pour les accumulateurs d’entrée et de sortie doivent être respectées.

4.3.5 Exemples

4.3.5.1 Sélection d’un élément par son indice

Comme nous l’avons dit plus haut, il est impossible en de sélectionner un élément d’un
tableau directement par son indice si celui-ci est donné par une expression dont la valeur n’est pas
connue statiquement.
Nous proposons, comme premier exemple d’utilisation des itérations, de coder un programme qui

sélectionne le i-ème élément d’un tableau array, la valeur de i étant donnée par une expression quel-
conque. Lorsque la valeur de i n’est pas valide (hors des bornes du tableau) le programme retourne
une valeur par défaut (une constante statique dans notre cas, mais on pourrait choisir de renvoyer

(c) mapfold

Figure 5: Iterators in Heptagon

Iterators can also be used for multidimensional arrays. In the case of mapi and foldi, the
iterated function should then expect one argument for each dimension.

It is also possible to mimic partial application with iterators, by giving a list of arguments,
not necessarily arrays, that will be given as first arguments to each application of the iterated
function. For instance, o =map<<n>>f <(a)>(t) is equivalent to o[i] =f(a, t[i]) for each
element in the output array o.

Arrays are semantically functional: each modification on an array creates a new array. Im-
plementing these arrays using separate arrays can lead to very unefficient code. That’s why an
optimization pass called memory allocation [1] tries to share arrays and avoid unnecessary copies
within each node. It can be enabled using the -memalloc or -O option.

2.4.2 Records

The syntax for declaring record types is the following:

type t = { f1: t1; f2: t2 }

Note that two different record types cannot have fields with the same name. It is possible to
declare a new record ({ f1 =x; f2 =9 }), read a field of a record (x.f1) and modify one field
(r’ ={ r with .f1 =v } returns a new record where all fields are equal to the ones in r except
for f1 which is equal to v).

2.4.3 Types alias

It is possible to declare an alias for any type (like a typedef in C), for instance:

type meter = int
type matrix = int^10^10

This alias can then be used anywhere a type is expected.

2.5 Parametricity
The size of arrays can be parametrized by so-called static expressions, that is expressions that
are reduced to constants at compile-time. They are either a literal, a global constant, a static
parameter of a node or a static operation on two static expressions (+, -, *, /, +., <=, or, etc).
Global constants are declared by:

const n : int = 100
const t0 : float^n = 1.0^n
const r0 = { f1 = 0; f2 = t0 }

The parameters of a node are given between << and >> at the declaration of the node and
are instantiated with the same syntax:

node f<<m:int; t1: int^n>>(a:int^m) = (o:int^m)
let
o = map<<m>> (+)(a, t1);
tel

node g(a:int^n) = (o:int^n)
let

o = f<<n, t0>>(a);
tel

If the backend support parametricity (like in Java), static parameters are kept in the generated
code. Otherwise, a pass of the compiler generates all the versions of the node that are needed.
If a parametrized node defined in a file f1.ept is used in f2.ept, it is necessary to first compile
f1.ept with the -c option (and without any -target), that generates a binary file f1.epo. The
compilation of the second file, this time with the -target option, will generate all the necessary
nodes, from f1.ept and f2.ept.

2.6 Location annotations
Memory allocation [1] avoids unnecessary array copies within the nodes automatically (it can
be enable with the -memalloc or -O options). In order to avoid copies when calling nodes, the
user must add location annotations. We will give here only a short introduction, the interested
reader can refer to [1] for more details. They express the fact that a function modifies in-place
its argument. For instance, consider this node that swaps two elements in an array:

node swap(i, j:int; t_in:float^100 at r) returns (t_out:float^100 at r)
var t_tmp:float^100 at r;
let
t_tmp = [ t_in with [i] = t_in[>j<] ];
t_out = [ t_tmp with [j] = t_in[>i<] ];
tel

The location annotations are introduced by the keyword at followed by a location name. All
the streams with the same location name are guaranteed to be stored together in the generated
code, so the function generated for swap will directly modify its argument in-place.

Located variables are called semilinear : they can only be updated once, but they can be read
many times. An update is a function that modify its argument in-place, for instance a node that
inputs and outputs located variables, and a read is any other function. For instance, modifying
one element in an array is an update, but accessing one element is a read.

Only located variables can be given to a function that expects located arguments. A located
variable can only be obtained by updating another located variable or by explicitly initializing a
new location with the init construction:

node shuffle(i_arr, j_arr : int^m; q : int) = (v : float)
var t, t_prev : float^n at r;
let
init<<r>> t_prev = t_0 fby t;
t = fold<<m>> swap(i_arr, j_arr, t_prev);
v = t[>q<];
tel

2.7 Interfaces
Interface files, with the extension .epi, do not contain any definition but only the declaration
(or signatures) of nodes, types and constants. In particular, they are useful to import external
functions. For instance:

external fun sort(a:int^100) = (o:int^100)

The imported function should respect the calling convention given in appendix A. See the direc-
tory examples/extern_C for a complete example.

3 BZR: Contracts for controller synthesis
Contracts are an extension of the Heptagon language, so as to allow to perform discrete controller
synthesis on Heptagon programs. The extended language is named BZR.

We associate to each node a contract, which is a program associated with two outputs :

• an output eA representing the environment model ;

• an invariance objective eG ;

• a set {c1, . . . , cn} of controllable variables used for ensuring this objective.

This contract means that the node will be controlled, i.e., that values will be given to c1, . . . , cn
such that, given any input trace yielding eA, the output trace will yield the true value for eG.

f(x1, . . . , xn) = (y1, . . . , yn)

assume eA

guarantee eG with (c1, . . . , cn)

y1 = f1(x1, . . . , xn, c1, . . . , cn)
. . .
yn = fn(x1, . . . , xn, c1, . . . , cn)

In the textual syntax, the contracts are noted :

node f(x1:t1;. . .;xn:tn) returns (y1:t′1;. . .;yp:t
′
p)

contract
var . . .
let

. . .
tel
assume eA
enforce eG
with (c1:t′′1;. . .;cq:t

′′
n)

var . . .
let
y1 = f1(x1, . . . , xn, c1, . . . , cq);
...
yp = fp(x1, . . . , xn, c1, . . . , cq);
tel

4 BZR Running Example: Multi-task System

4.1 Delayable Tasks
We consider a multi-task system composed of n delayable tasks. Figure 6 shows a delayable task.
A delayable task takes three inputs r, c and e: r is the task launch request from the environment,
e is the end request, and c is meant to be a controllable input controlling whether, on request,
the task is actually launched (and therefore goes in the active state), or delayed (and then forced
by the controller to go in the waiting state by stating the false value to c). This node outputs a
unique boolean act which is true when the task is in the active state.

node delayable(r,c,e:bool) returns (act:bool)
let
automaton
state Idle
do act = false
until r & c then Active
| a & not c then Wait

state Wait
do act = false
until c then Active
state Active
do act = true
until e then Idle

end
tel

Figure 6: Delayable task

The Figure 7 shows then a node ntasks where n delayable tasks have been put in parallel.
The tasks are inlined so as to be able to perform DSC on this node, taking into account the
tasks’ states. Until now, the only interest of modularity is, from the programmer’s point of view,
to be able to give once the delayable task code.

This ntasks node is provided with a contract, stating that its composing tasks are exclusive,
i.e., that there are no two tasks in the active state at the same instant. This contract is enforced
with the help of the controllable inputs ci.

4.2 Contract composition
We want know to reuse the ntasks node, in order to build modularly a system composed of 2n
tasks. The Figure 8 shows the parallel composition of two ntasks nodes. We associate to this
composition a new contract, which role is to enforce the exclusivity of the 2n tasks.

It is easy to see that the contract of ntasks is not precise enough to be able to compose several
of these nodes. Therefore, we need to refine this contract by adding some way to externally control
the activity of the tasks.

4.3 Contract refinement
We first add an input c, meant to be controllable. The refined contract will enforce that:

node ntasks(r1, . . . , rn, e1, . . . , en:bool)
returns (a1, . . . , an:bool)

contract
let
ca1 = a1 & (a2 or . . . or an);
...
can−1 = an−1 & an;
tel
enforce not (ca1 or \ldots or can−1)
with (c1, . . . , cn:bool)
let
a1 = inlined delayable(r1,c1,e1);
...
an = inlined delayable(rn,cn,en);
tel

Figure 7: ntasks node: n delayable tasks in parallel

1. the tasks are exclusive,

2. one task is active only at instants when the input c is true. This property, appearing
in the contract, allow a node instantiating ntasks to forbid any activity of the n tasks
instantiated.

The Figure 9 contains this new ntasks node.
However, the controllability introduced here is know too strong. The synthesis will succeed,

but the computed controller, without knowing how c will be instantiated, will actually block
every tasks in their idle state. Indeed, if the controller allows one task to go in its active state,
the input c can become false at the next instant, violating the property to enforce.

Thus, we propose to add an assumption to this contract: the input c will not become false if
a task was active an instant before. This new contract is visible in Figure 10.

We can then use this new ntasks version for the parallel composition, by instantiating the c
input by a controllable variable and its negation. This composition can be found in Figure 11.

A Generated code

A.1 C generated code
C generated files from an Heptagon program example.ept are placed in a directory named
example_c. This directory contains one file example.c. For each node f of the source program,
assuming that f has inputs (x1 : t1, . . . , xn : tn) and outputs (y1 : t′1, . . . , yp : t′p), ti and t′i being
the data types of these inputs and outputs, then the example.c file contains, for each node f:

• A Example__f_reset function, with an argument self being a memory structure instance:

void Example__f_reset(Example__f_mem* self);

15



node main(r1, . . . , r2n, e1, . . . , e2n:bool)
returns (a1, . . . , a2n:bool)

contract
let
ca1 = a1 & (a2 or . . . or a2n);
...
ca2n−1 = a2n−1 & a2n;
tel
enforce not (ca1 or . . . or ca2n−1)
let
(a1, . . . , an) = ntasks(r1, . . . , rn,e1, . . . , en);
(an+1, . . . , a2n) = ntasks(rn+1, . . . , r2n,en+1, . . . , e2n);
tel

Figure 8: Composition of two ntasks nodes

• A Example__f_step function, with as arguments the nodes inputs, a structure _out where
the output will be put, and a memory structure instance self:

void Example__f_step(t1 x1, ..., tn xn,
Example__f_out* \_out,
Example__f_mem* self);

After the call of this function, the structure _out contains the outputs of the node:

typedef struct \{
t′1 y1;
...
t′p yp;
\} Example__f_ans;

An example of main C code for the execution of this node would be then:

#include "example.h"

int main(int argc, char * argv[]) \{

Example__f_m mem;
t1 x1;
...
tn xn;
Example__f_out ans;

/* initialize memory instance */
f_reset(&mem);

while(1) \{
/* read inputs */

node ntasks(c,r1, . . . , rn,e1, . . . , en:bool) returns (a1, . . . , an:bool)
contract
let
ca1 = a1 & (a2 or . . . or an);. . .
can−1 = an−1 & an;
one = a1 or . . . or an;
tel
enforce not (ca1 or . . . or can−1) & (c or not one)
with (c1, . . . , cn:bool)
let
a1 = inlined delayable(r1,c1,e1);
...
an = inlined delayable(rn,cn,en);
tel

Figure 9: First contract refinement for the ntasks node

scanf("...", &x1, ..., &xn);

/* perform step */
Example__f_step(x1, ..., xn, &ans, &mem);

/* write outputs */
printf("...", ans.y1, ..., ans.yp);
\}
\}

The above code is nearly what is produce for the simulator with the -s option (see Section 1.4).

A.2 Java generated code
Java generated files from an Heptagon program example.ept are placed in a directory named
example_java. This directory contains one Java class f (in the file f.java) for each node f of the
source program. Assuming that f has inputs (x1 : t1, . . . , xn : tn) and outputs (y1 : t′1, . . . , yp :
t′p), ti and t′i being the data types of these inputs and outputs, then this f class implements the
following interface:

public interface f {

public void reset();

public fAnswer step(t1 x1, ..., tn xn);
}

The fAnswer class being a structure containing the outputs:

public class fAnswer {
t′1 y1;
...

node ntasks(c,r1, . . . , rn,e1, . . . , en:bool) returns (a1, . . . , an:bool)
contract
let
ca1 = a1 & (a2 or . . . or an);. . .
can−1 = an−1 & an;
one = a1 or . . . or an;
pone = false fby one;
tel
assume (not pone or c)
enforce not (ca1 or . . . or can−1) & (c or not one)
with (c1, . . . , cn)
let
a1 = inlined delayable(r1,c1,e1);
...
an = inlined delayable(rn,cn,en);
tel

Figure 10: Second contract refinement for the ntasks node

t′p yp;
}

node main(r1, . . . , r2n,e1, . . . , e2n:bool) returns (a1, . . . , a2n:bool)
contract
let
ca1 = a1 & (a2 or . . . or a2n);
...
ca2n−1 = a2n−1 & a2n;
tel
enforce not (ca1 or . . . or ca2n−1)
with (c:bool)
let
(a1, . . . , an) = ntasks(c,r1, . . . , rn,e1, . . . , en);
(an+1, . . . , a2n) = ntasks(\Not c,rn+1, . . . , r2n,en+1, . . . , e2n);
tel

Figure 11: Two ntasks parallel composition

